HRP 261 SAS LAB SIX, February 22, 2012

Lab Six: Data exploration exercise: model building for logistic regression

Lab Objectives

After today’s lab you should be able to:

1. Explore a dataset to identify outliers and missing data.

2. Plot data distributions. 

3. Obtain Pearson’s correlation coefficients between multiple covariates (must be continuous or binary).

4. Build a logistic regression model for hypothesis-driven research.

5. Generate ORs for categorical variables and specify a reference group.

6. Look for confounding and effect modification in the context of logistic regression.

7. Understand the contrast statement.

8. Walk through a real data analysis exercise.

 LAB EXERCISE STEPS:

Follow along with the computer in front…

1. Download the LAB 6 DATA (chd) from the class website (already in SAS format!).

www.stanford.edu/~kcobb/courses/hrp261 (right click on LAB 6 DATA(save to desktop

This dataset contains data from an unmatched case-control study of 160 chd cases (heart disease) and 302 controls. Participants were queried about their medical status and personal habits one year ago (prior to the onset of heart disease for cases). 

Outcome variable:

CHD—heart disease, yes/no (1/0)

Predictor variables:

Age—years 

Tobacco—cigarettes/day

Alcohol—ounces/day

Adiposity—percent body fat

BMI—normal weight, overweight, or obese according to BMI (character variable)

Sbp—blood pressure

LDL—LDL cholesterol

FamHist—Family history of heart disease (1/0)

Typea—Score on a test of type A personality (higher score means more Type A)

The purpose of the study was to test whether alcohol and tobacco are related to heart disease controlling for potential confounders.

2. Use point-and-click features to create a permanent library that points to the desktop (where the dataset is sitting):

a. Click on Tools>Assign Project Library (slamming file cabinet)

b. Name the library lab6.

c. Click on Browse and select your desktop

 (This is to create the library in your desktop )

d. Click next and finish to create the library in your desktop.
You can also name a library using code:

LIBNAME LAB6 "C:\Your Desktop Path”;  

3. Click on View>Process flow to see the process flow window. Place your mouse in front of the CHD icon to check that it is saved on the correct folder

4. Use the distribution analysis tool to check the variables in the dataset chd:
a. From the menu on the dataset select: Describe(Distribution Analysis

b. Select “sbp” variable as Analysis variables

c. Select Plot>Histogram 

d. Select Tables>Basic confidence intervals, Basic Measures, Tests of location, Moments, Quartiles. 

e. Repeat for the other variables.

f. What things do you notice?

g. What’s your sample size? How many men have chd? (Hint: Click on Describe>One-Way Frequencies(Chd)
h. What variables are correlated with chd? Click on Analyze>Multivariate>Correlations. Then select sbp, tobacco, ldl, adiposity, famhist, typea, alcohol, age as Analysis Variables and chd as Correlate with

On the Results menu check the box Show correlations in decreasing order of magnitude. Then check the box Show n correlations per row variable and select n = 5.

Corresponding code would be:
proc corr data=lab6.chd best=5;

var sbp tobacco ldl adiposity famhist typea 


alcohol age;

run;


	Pearson Correlation Coefficients, N = 462 
Prob > |r| under H0: Rho=0

	tobacco

tobacco


	tobacco

1.00000

 


	age

0.45033

<.0001


	adiposity

0.28664

<.0001


	sbp

0.21225

<.0001


	alcohol

0.20081

<.0001



	ldl

ldl


	ldl

1.00000

 


	adiposity

0.44043

<.0001


	age

0.31180

<.0001


	famhist

0.16135

0.0005


	tobacco

0.15891

0.0006



	adiposity

adiposity


	adiposity

1.00000

 


	age

0.62595

<.0001


	ldl

0.44043

<.0001


	sbp

0.35650

<.0001


	tobacco

0.28664

<.0001



	famhist

famhist


	famhist

1.00000

 


	age

0.23967

<.0001


	adiposity

0.18172

<.0001


	ldl

0.16135

0.0005


	tobacco

0.08860

0.0570



	typea

typea


	typea

1.00000

 


	age

-0.10261

0.0274


	sbp

-0.05745

0.2177


	famhist

0.04481

0.3366


	ldl

0.04405

0.3448



	alcohol

alcohol


	alcohol

1.00000

 


	tobacco

0.20081

<.0001


	sbp

0.14010

0.0025


	age

0.10112

0.0298


	adiposity

0.10033

0.0311



	age

age


	age

1.00000

 


	adiposity

0.62595

<.0001


	tobacco

0.45033

<.0001


	sbp

0.38877

<.0001


	ldl

0.31180

<.0001



	sbp

sbp


	sbp

1.00000

 


	age

0.38877

<.0001


	adiposity

0.35650

<.0001


	tobacco

0.21225

<.0001


	ldl

0.15830

0.0006




The high correlations among the variables means that we may have issues with co-linearity and confounding. 

5. Run the full logistic model to see the effects of co-linearity:

Run the model:

 Logit(chd)  = sbp + tobacco + ldl + adiposity + famhist + types + bmi + alcohol + age

*Don’t forget to fit your model to the value “1” rather than 0. 

*Don’t forget to select all variables as main effects.

*Select bmi with your mouse and select Coding Style for bmi>Reference. Then modify the code to make normal the reference category for bmi:

class bmi (ref="normal" param=ref);

	Analysis of Maximum Likelihood Estimates

	Parameter
	 
	DF
	Estimate
	Standard
Error
	Wald
Chi-Square
	Pr > ChiSq

	Intercept
	 
	1
	-7.2912
	1.1855
	37.8269
	<.0001

	sbp
	 
	1
	0.00636
	0.00571
	1.2385
	0.2658

	tobacco
	 
	1
	0.0799
	0.0265
	9.0953
	0.0026

	ldl
	 
	1
	0.1790
	0.0604
	8.7893
	0.0030

	adiposity
	 
	1
	0.00413
	0.0272
	0.0229
	0.8796

	famhist
	 
	1
	0.9223
	0.2278
	16.3958
	<.0001

	typea
	 
	1
	0.0383
	0.0124
	9.5856
	0.0020

	bmi
	obese
	1
	-0.3808
	0.4739
	0.6455
	0.4217

	bmi
	overwe
	1
	-0.2922
	0.3163
	0.8536
	0.3555

	alcohol
	 
	1
	0.000584
	0.00449
	0.0169
	0.8966

	age
	 
	1
	0.0477
	0.0121
	15.5027
	<.0001


Corresponding code to perform this analysis would be:
proc logistic data=lab6.chd;

class bmi (ref="normal" param=ref);

model chd (event="1") = sbp tobacco ldl adiposity famhist typea 


bmi alcohol age /risklimits;

run;

6. Run a new model with only one predictor using code (New Program):
Logit(chd) = bmi
proc logistic data=lab6.chd;

class bmi (ref="normal" param=ref);

model chd (event="1") = bmi /risklimits;

run;
	Analysis of Maximum Likelihood Estimates

	Parameter
	 
	DF
	Estimate
	Standard
Error
	Wald
Chi-Square
	Pr > ChiSq

	Intercept
	 
	1
	-0.9444
	0.1575
	35.9623
	<.0001

	bmi
	obese
	1
	0.6407
	0.2844
	5.0767
	0.0242

	bmi
	overwe
	1
	0.4813
	0.2171
	4.9156
	0.0266




	Wald Confidence Interval for Odds Ratios

	Effect
	Unit
	Estimate
	95% Confidence Limits

	bmi obese vs normal
	1.0000
	1.898
	1.087
	3.314

	bmi overwe vs normal
	1.0000
	1.618
	1.057
	2.476


7. Just for kicks, let’s try using automatic selection procedures (stepwise, forward, backward) to select the predictors for our model. Automatic selection procedures are essentially fishing expeditions, and should only be used for exploratory purposes. They may miss important predictors, confounders, and effect modifiers, and may throw out your most important predictor of interest.  

They are not recommended in the context of hypothesis-driven research! But can be used when you are exploring your data, prior to formal model-building. 
a) **Double click on the Logistic Regression point-and-click icon**. Then select Modify Task

Select Model>Selection>Stepwise Selection

b) Run the model with forward selection.
Model>Selection>Forward Selection

Final model is:

	Analysis of Maximum Likelihood Estimates

	Parameter
	DF
	Estimate
	Standard
Error
	Wald
Chi-Square
	Pr > ChiSq

	Intercept
	1
	-6.4464
	0.9209
	49.0051
	<.0001

	age
	1
	0.0505
	0.0102
	24.4446
	<.0001

	tobacco
	1
	0.0804
	0.0259
	9.6456
	0.0019

	ldl
	1
	0.1620
	0.0550
	8.6846
	0.0032

	famhist
	1
	0.9082
	0.2258
	16.1827
	<.0001

	typea
	1
	0.0371
	0.0122
	9.3058
	0.0023


c) Run the model with stepwise selection. **Double click on the Logistic Regression point-and-click icon**. Then select Modify Task
Model>Selection>Stepwise Selection

Picks out the same predictors as forward selection (the most statistically significant predictors):

	Analysis of Maximum Likelihood Estimates

	Parameter
	DF
	Estimate
	Standard
Error
	Wald
Chi-Square
	Pr > ChiSq

	Intercept
	1
	-6.4464
	0.9209
	49.0051
	<.0001

	age
	1
	0.0505
	0.0102
	24.4446
	<.0001

	tobacco
	1
	0.0804
	0.0259
	9.6456
	0.0019

	ldl
	1
	0.1620
	0.0550
	8.6846
	0.0032

	famhist
	1
	0.9082
	0.2258
	16.1827
	<.0001

	typea
	1
	0.0371
	0.0122
	9.3058
	0.0023


Note that all result in the same model, which includes: tobacco, ldl, famhist, typea, age. This simply confirms that these are strong predictors.


Corresponding code would be:
proc logistic data = lab6.chd;

class bmi (ref="normal" param=ref);

model chd (event="1") = sbp tobacco ldl bmi

adiposity famhist typea alcohol age 

/selection=stepwise sle=.05 sls=.05;

run;

proc logistic data = lab6.chd;

class bmi (ref="normal" param=ref);

model chd (event="1") = sbp tobacco ldl bmi

adiposity famhist typea alcohol age  

/selection=forward sle=.05;

run;

proc logistic data = lab6.chd;

class bmi (ref="normal" param=ref);

model chd (event="1") = sbp tobacco ldl bmi

adiposity famhist typea alcohol age  

/selection=backward sls=.05;

run;


8. Test the hypothesis that tobacco is related to chd, starting with a univariate regression:
a) Analyze(Regression(logistic regression
b) Model: logit (chd) = tobacco
c) Select Model>Response>Fit Model to level “1”
d) Select tobacco as a main effect. 

The corresponding code is the following:

proc logistic data = lab6.chd;

model chd (event="1") = tobacco; 
run;
	run;

Analysis of Maximum Likelihood Estimates

	Parameter
	DF
	Estimate
	Standard
Error
	Wald
Chi-Square
	Pr > ChiSq

	Intercept
	1
	-1.1894
	0.1390
	73.2280
	<.0001

	tobacco
	1
	0.1453
	0.0248
	34.4098
	<.0001


	Odds Ratio Estimates

	Effect
	Point Estimate
	95% Wald
Confidence Limits

	tobacco
	1.156
	1.102
	1.214


Univariate model:

Logit(chd) = -1.1894 + 0.1453 (cigarettes/day)

Interpretation: every 1 cigarette smoked per day increases odds of chd by 15.6% (statistically significant). (10 cigarettes would translate to exp(1.453)= 4.257, 325.7% increase in risk).

Question: Does it make sense to model cigarettes as a continuous predictor? Does every increase in 1 cigarette per day really increase risk by 15%, or do the first few cigarettes have a bigger influence, with diminishing returns? 

To save time, I already ran logit plots, using the logit plot macro (try this on your own later):

Example code: 

%logitplot(lab6.chd, tobacco, chd, NumBins= 4);

Here’s what it looks like

4 bins:





10 bins:

As expected, it’s not linear in the logit. Rather, it looks a little more curvilinear, with an initial big jump going from nonsmoker (0 cigarettes per day) to smoker (even 1 cigarette/day), and then smaller jumps in risk with more cigarettes smoked. 

We can also look at the relationship between chd and number of cigarettes in smokers only:

4 bins:





10 bins:

For smokers, it appears that there’s two peaks of risk—one for lighter smokers and one for heavier smokers. 
Potentially then, we should model smoking as a categorical variable—with categories of nonsmoker, light smoker, and heavy smoker. 

10. Try modeling non-smoker vs. light smoker vs. heavy smoker, using “clinical” definitions of light and heavy smoking:

a) Write the following code to define a new categorical variable called smoker and add it to the dataset:
data chd;

set lab6.chd;

if tobacco=0 then smoker="non";

else if 3>=tobacco>0 then smoker="light";

else if tobacco>3 then smoker="heavy";

run;

b) Using code, run the logistic model logit(chd) = smoker. Set “non” as the reference category on smoker:
proc logistic data = chd;

class smoker (ref="non" param=ref);

model chd (event="1") = smoker;  

run;

	Analysis of Maximum Likelihood Estimates

	Parameter
	 
	DF
	Estimate
	Standard
Error
	Wald
Chi-Square
	Pr > ChiSq

	Intercept
	 
	1
	-1.8136
	0.2784
	42.4242
	<.0001

	smoker
	hea
	1
	1.7719
	0.3136
	31.9164
	<.0001

	smoker
	lig
	1
	1.0269
	0.3257
	9.9421
	0.0016


	Odds Ratio Estimates

	Effect
	Point Estimate
	95% Wald
Confidence Limits

	smoker hea vs non
	5.882
	3.181
	10.876

	smoker lig vs non
	2.792
	1.475
	5.287


Interpretation: heavy smokers have a 6-fold increase in their odds of chd compared with non-smokers; light smokers have a nearly 3-fold increase. 

11. Test for confounding by the potential confounders. Parameter estimates are 1.77 for heavy smokers; 1.02 for light smokers.

To test for confounding, include each covariate in the model one at a time with the main predictor and see how inclusion of the covariate affects the relationship between tobacco and the outcome (i.e., affects the beta coefficient). To speed up the process of doing this, use a MACRO:


%macro speed(confounder);

proc logistic data = chd;

class smoker (ref="non" param=ref);

model chd (event="1") = smoker &confounder.;  

run;

%mend;

%speed(alcohol);

%speed(adiposity);

%speed(ldl);

%speed(typea);

%speed(famhist);

%speed(sbp);

%speed(age);
Scroll through the output to find that there is strong confounding by age. Age is strongly related to smoking, and age is strongly related to chd. Borderline: LDL cholesterol. 

Also, must run a separate model for the categorical potential confounder, bmi:

proc logistic data = chd;

class smoker (ref="non" param=ref) bmi (ref="normal" param=ref);

model chd (event="1") = smoker bmi;

run;
No evidence of confounding by bmi class.

12. Next check for effect modification. Generally, you would only want to test for interactions that you had specified a priori, or that had some biological rationale. For the purposes of this example, let’s say we were most interested in potential interactions with alcohol and adiposity/BMI:
%macro speed(interact);

proc logistic data = chd;

class smoker (ref="non" param=ref);

model chd (event="1") = smoker &interact.
smoker*&interact.;  

run;

%mend;

%speed(alcohol);

%speed(adiposity);

Scroll through the output to find that there is no evidence of interaction. Neither of the interaction terms approach statistical significance despite having a large sample size here.

Run separate model for bmi class:

proc logistic data = chd;

class smoker (ref="non" param=ref) bmi (ref="normal" param=ref);

model chd (event="1") = smoker bmi smoker*bmi;

run;
No significant interactions.

13. Now, assemble your final model using point and click (use the units option to get ORs in terms of 10-years of age and 10-points of typea personality):

a) Analyze(Regression(Logistic Regression

b) Logit (chd) = smoker + age + ldl + famhist + typea 
c) Set the units of age and typea to 10

d) Select Model>Response>Fit Model to 1
e) Select all variables as main effects

f) Set “non” as the reference class for smoker 

class smoker (ref="non" param=ref);

Corresponding code:

proc logistic data = chd;

class smoker (ref="non" param=ref);

model chd (event="1") = smoker age ldl famhist typea;

units age=10 typea=10;

run;

	Analysis of Maximum Likelihood Estimates

	Parameter
	 
	DF
	Estimate
	Standard
Error
	Wald
Chi-Square
	Pr > ChiSq

	Intercept
	 
	1
	-6.9207
	0.9492
	53.1578
	<.0001

	age
	 
	1
	0.0550
	0.0102
	29.2533
	<.0001

	ldl
	 
	1
	0.1492
	0.0546
	7.4783
	0.0062

	famhist
	 
	1
	0.8611
	0.2240
	14.7766
	0.0001

	typea
	 
	1
	0.0367
	0.0120
	9.3078
	0.0023

	smoker
	hea
	1
	0.9159
	0.3539
	6.6965
	0.0097

	smoker
	lig
	1
	0.7453
	0.3621
	4.2357
	0.0396


	Odds Ratio Estimates

	Effect
	Point Estimate
	95% Wald
Confidence Limits

	age
	1.057
	1.036
	1.078

	ldl
	1.161
	1.043
	1.292

	famhist
	2.366
	1.525
	3.670

	typea
	1.037
	1.013
	1.062

	smoker hea vs non
	2.499
	1.249
	5.001

	smoker lig vs non
	2.107
	1.036
	4.285


	Odds Ratios

	Effect
	Unit
	Estimate

	age
	10.0000
	1.733

	typea
	10.0000
	1.443


(Note: If your goal is JUST to report adjusted and unadjusted ORs for smoking, then inclusion of famhist and typea is not necessary. Taking them out of the model does not affect OR’s for smoking. But you may want to report additional chd predictors as part of your secondary analyses.)

Reporting the results: Report as ORs:
Unadjusted ORs (and 95% confidence intervals):

Heavy smoker:  
5.88 (3.18, 10.88)
Light smoker: 

2.79  (1.48, 5.29)
Non-smoker (reference): 1.00 (reference)

Adjusted* ORs (and 95% confidence intervals):

Heavy smoker:  



2.50 (1.25, 5.00)

Light smoker: 




2.11 (1.04, 4.29)

Non-smoker (reference): 


1.00 (reference)

Age (per 10 year increase):  


1.73 (1.42, 2.12)

LDL cholesterol (per mg/ml increase): 
1.16 (1.04, 1.29)
Family history of heart disease: 

1.04 (1.01, 1.06)
Type A personality (per 10-point increase):  
1.44 (1.14, 1.83)
*Adjusted for all other predictors in the table.
14. What if you want to get the OR and 95% CI for heavy smoker versus light smoker?  You could just re-run the logistic regression with a new reference group (light). Or, you can specify contrasts as follows:

proc logistic data = chd;

class smoker (ref="non" param=ref);

model chd (event="1") = smoker age famhist

typea ldl ;  

contrast 'heavy vs. light' smoker  1 -1 0 /estimate=exp;

run;

	Contrast Test Results

	Contrast
	DF
	Wald
Chi-Square
	Pr > ChiSq

	heavy vs. light
	1
	0.4582
	0.4985


	Contrast Rows Estimation and Testing Results

	Contrast
	Type
	Row
	Estimate
	Standard
Error
	Alpha
	Confidence Limits
	Wald
Chi-Square
	Pr > ChiSq

	heavy vs. light
	EXP
	1
	1.1860
	0.2989
	0.05
	0.7237
	1.9436
	0.4582
	0.4985


OR and 95% CI=1.1860 (.7237, 1.9436), p=.4985
15. Now that you have your final model, get individual predictions and residuals:

a) **Double click on the logistic regression icon** Select Modify Task. 
b) On the Predictions menu select the boxes 
Data to Predict>Original Samples 

Additional Statistics>Residuals, Prediction limits
Save output data>Predictions

Display output and plots
Display output and Plots>Show predictions

c) On the menu Predictions click on browse and call the output out. Run
d) To print your results Click on Modify Task and select the box Predictions>Show predictions
	Analysis of Maximum Likelihood Estimates

	Parameter
	 
	DF
	Estimate
	Standard
Error
	Wald
Chi-Square
	Pr > ChiSq

	Intercept
	 
	1
	-6.9207
	0.9492
	53.1578
	<.0001

	smoker
	hea
	1
	0.9159
	0.3539
	6.6965
	0.0097

	smoker
	lig
	1
	0.7453
	0.3621
	4.2357
	0.0396

	age
	 
	1
	0.0550
	0.0102
	29.2533
	<.0001

	ldl
	 
	1
	0.1492
	0.0546
	7.4783
	0.0062

	famhist
	 
	1
	0.8611
	0.2240
	14.7766
	0.0001

	typea
	 
	1
	0.0367
	0.0120
	9.3078
	0.0023


Final model:

SAS transforms the residual into a Z-score (deviance residual): here 1.19.

Code:
proc logistic data = chd;

        class smoker (ref="non" param=ref);

model chd (event="1") = smoker age ldl famhist typea;
output out = outdata l = Lower p = Predicted u = Upper resdev=residuals;

run;

proc print data=outdata;

var smoker age ldl famhist typea chd predicted residuals;

run;

OPTIONAL FURTHER PRACTICE:
You can also test the relationship between alcohol and chd:

a) Run the logistic model Logit (chd) = alcohol

	Analysis of Maximum Likelihood Estimates

	Parameter
	DF
	Estimate
	Standard
Error
	Wald
Chi-Square
	Pr > ChiSq

	Intercept
	1
	-0.7260
	0.1200
	36.6006
	<.0001

	alcohol
	1
	0.00520
	0.00389
	1.7840
	0.1817

	Odds Ratio Estimates

	Effect
	Point Estimate
	95% Wald
Confidence Limits

	alcohol
	1.005
	0.998
	1.013


Code:
proc logistic data = lab6.chd;

model chd (event="1") = alcohol;

run;
No evidence of a relationship at first glance.  However, would we really expect alcohol to be linear in the logit?  Maybe not—moderate alcohol drinking has been related to protection against cardiovascular disease in the past, whereas heavy drinking has been related to increased risk. So, we are not expecting a linear relationship.

To save time, I already ran logit plots (try this on your own later):

%logitplot(lab6.chd, alcohol, chd, NumBins=4);

This is very interesting!  Note the steep drop in risk at light to moderate alcohol drinking (approximately 1 drink/day), but the sharp increase in risk thereafter.  This suggests that our failure to find an association may be due to the lack of a linear relationship (rather than the lack of any relationship).

16. Model alcohol as categorical (using clinically meaningful categories): non-drinkers (0 drinks/day), light drinker (up to 1 drink/day), moderate drinker (1 to 3 drinks per day), and heavy drinkers (>3 drinks per day); 1 drink= 4 ounces:

data chd;

set lab6.chd;

if alcohol=0 then drinker="non";

else if alcohol<=4 then drinker="light";

else if alcohol<=12 then drinker="moderate";

else if alcohol>12 then drinker="heavy";

run;

proc logistic data = chd;

class drinker (ref="non" param=ref);

model chd (event="1") = drinker;  

run;

	Analysis of Maximum Likelihood Estimates

	Parameter
	 
	DF
	Estimate
	Standard
Error
	Wald
Chi-Square
	Pr > ChiSq

	Intercept
	 
	1
	-0.6795
	0.2018
	11.3387
	0.0008

	drinker
	hea
	1
	0.2605
	0.2519
	1.0696
	0.3010

	drinker
	lig
	1
	-0.3592
	0.3156
	1.2954
	0.2551

	drinker
	mod
	1
	0.00509
	0.3109
	0.0003
	0.9869


	Odds Ratio Estimates

	Effect
	Point Estimate
	95% Wald
Confidence Limits

	drinker hea vs non
	1.298
	0.792
	2.126

	drinker lig vs non
	0.698
	0.376
	1.296

	drinker mod vs non
	1.005
	0.547
	1.848


Note the reduced risk for light drinkers and the increased risk for heavy drinkers, though nothing is statistically significant.

17. It’s possible that there is a confounder that is masking the effect of alcohol. Test for confounding. Unadjusted parameter estimates are: 0.2605, -.3592, and .00509 for heavy, light, and moderate, respectively.

%macro speed(confounder);

proc logistic data = chd;

class drinker (ref="non" param=ref);

model chd (event="1") = drinker &confounder.;  

run;

%mend;

%speed(tobacco);

%speed(adiposity);

%speed(ldl);

%speed(typea);

%speed(famhist);

%speed(sbp);

%speed(age);
Age and tobacco appear to be the major confounders. Heavy drinkers are heavy smokers, and that completely explains their increased risk of chd. The drinking groups also appear more similar when age is included in the model, suggesting that heavier drinkers are older, which partly explains their increased chd risk. Note that none of the alcohol-chd relationships becomes stronger when confounders are included; the relationships are only attenuated.

18. Final model (if you want to also report the other significant predictors of chd):

proc logistic data = chd;

class drinker (ref="non" param=ref) bmi (ref="normal" param=ref);

model chd (event="1") = drinker age tobacco famhist

typea ldl ;  

run;

	Analysis of Maximum Likelihood Estimates

	Parameter
	 
	DF
	Estimate
	Standard
Error
	Wald
Chi-Square
	Pr > ChiSq

	Intercept
	 
	1
	-6.4183
	0.9534
	45.3251
	<.0001

	drinker
	hea
	1
	0.0609
	0.2924
	0.0434
	0.8349

	drinker
	lig
	1
	-0.2614
	0.3653
	0.5121
	0.4742

	drinker
	mod
	1
	-0.0782
	0.3619
	0.0467
	0.8290

	age
	 
	1
	0.0505
	0.0103
	24.1467
	<.0001

	tobacco
	 
	1
	0.0767
	0.0263
	8.5264
	0.0035

	famhist
	 
	1
	0.9098
	0.2262
	16.1782
	<.0001

	typea
	 
	1
	0.0371
	0.0122
	9.2678
	0.0023

	ldl
	 
	1
	0.1658
	0.0552
	9.0196
	0.0027


	Odds Ratio Estimates

	Effect
	Point Estimate
	95% Wald
Confidence Limits

	drinker hea vs non
	1.063
	0.599
	1.885

	drinker lig vs non
	0.770
	0.376
	1.576

	drinker mod vs non
	0.925
	0.455
	1.880

	age
	1.052
	1.031
	1.073

	tobacco
	1.080
	1.026
	1.137

	famhist
	2.484
	1.594
	3.869

	typea
	1.038
	1.013
	1.063

	ldl
	1.180
	1.059
	1.315


19. If you want to make any other pairwise comparisons besides heavy vs. non-drinker and light vs. non-drinker etc., for example if you wanted to compare heavy to light drinkers or heavy to moderate, either change the reference group or ask for contrasts:

proc logistic data = chd;

class drinker (ref="non" param=ref) bmi (ref="normal" param=ref);

model chd (event="1") = drinker age tobacco famhist

typea ldl ;  

contrast 'heavy vs. light' drinker  1 -1 0 0 /estimate=exp;

contrast 'heavy vs. moderate' drinker 1 0 -1 0 /estimate=exp;

run;

	Contrast Test Results

	Contrast
	DF
	Wald
Chi-Square
	Pr > ChiSq

	heavy vs. light
	1
	0.9581
	0.3277

	heavy vs. moderate
	1
	0.1827
	0.6691


	Contrast Rows Estimation and Testing Results

	Contrast
	Type
	Row
	Estimate
	Standard
Error
	Alpha
	Confidence Limits
	Wald
Chi-Square
	Pr > ChiSq

	heavy vs. light
	EXP
	1
	1.3804
	0.4546
	0.05
	0.7239
	2.6323
	0.9581
	0.3277

	heavy vs. moderate
	EXP
	1
	1.1492
	0.3740
	0.05
	0.6073
	2.1747
	0.1827
	0.6691









On univariate analysis, obesity and overweight are significantly harmful.





Funny result of obesity  and overweight appearing slightly protective! 


Why?


BMI is strongly correlated to adiposity, which is also in the model.


























heavy/light/non (ALPHABETICAL ORDER!!)


1         0        -1   compares heavy to non-smoker


0         1        -1   compares light to non-smoker


1        -1         0   compares heavy to light smoker





OR heavy vs. light = 1.38 (.72, 2.6)


OR heavy vs. moderate = 1.15 (.61, 2.17)











Stepwise selection enters variables into the model and removes them based on a pre-set significance level.





Forward selection enters variables into the model one at a time based on a pre-set significance level.





Backward selection removes  variables from the model one at a time based on a pre-set significance level.





All three methods try to get at the best combination of predictors (in terms of prediction), but they’re completely based on p-values, not intelligence. They’re likely to capture the strongest predictors well, but miss confounding, interaction, weaker predictors, and main predictors of interest.











Use sle (significance level entry) option to change significance criterion for entry into the model (default=.05) for stepwise and forward selection.





Use sls option to change significance criterion for removal from the model (default=.05) for stepwise and backward selection.

















Name the contrast something informative.





Variable name (must be categorical).





Name the contrast something informative.





PROC CORR gives Pearson correlation coefficients (linear correlations) between variables on the var line (binary or continuous variables only).








Variable name (must be categorical).





Asks for ORs and 95% CIs





Asks for ORs and 95% CIs





heavy /light /moderate /non (Alphabetical!)


1        -1           0            0   compares heavy to light


1         0           -1           0   compares heavy to moderate











To limit the output, the “best=k” option asks for the k highest correlation coefficients for each variable (including its correlation with itself).








Recall:” &variable. is a macro variable.

